

# A formal account of Archi deponent nouns: deponency without defectiveness and deponency involving heteroclisis

Andrew Hippisley

## 1 Introduction

Archi deponent nouns involve the mismatch between morphosyntactic number and its formal expression. It has four main properties, each of which needs to be addressed in a formal treatment. First, defectiveness in the paradigm does not accompany deponency as it does in Latin. This can be explained by a further three properties, as we shall see. The second property is that deponency may involve heteroclisis. Third, number is marked on the stem rather than the inflection, therefore the mismatch involves the stem. And fourth, Archi nouns may have a version of the root associated with a particular number, seen in cases of weak and strong suppletion. In this event, there is no mismatch between morphosyntactic number and its formal expression at the level of the root, even in a deponent noun. We offer a formal account within the Network Morphology framework (Corbett and Fraser 1993; Evans, Brown and Corbett 2001; Hippisley 2001) that captures all these salient facts represented in the lexical knowledge representation language DATR (Evans and Gazdar 1996). We begin by discussing our formal treatment of the noun system in terms of a system of stem formation classes (section 2). We then show how lexical entries expressing regular nouns (section 3) and deponent nouns (section 4) inherit the relevant facts. Finally, for completeness we incorporate strong suppletive nouns, following the stem-based approach discussed in Hippisley, Chumakina, Corbett and Brown (2004). The analysis of the Archi data is taken from [website of Matthew's language report] which in turn draws on Kibrik (1977a; 1977b).

## 2 The Noun System in Archi

Archi nouns have distinct singular and plural stems on which are built the case inflectional markers. Number distinctions are therefore expressed by the stem. Oblique case markers are based on the ergative word-form. There are no inflectional classes as such: all nouns that can inflect for a morphosyntactic feature inflect in the same way. However there are three main stem formation classes that provide suitable number distinguishing stems for inflection. Regardless of class, by default the absolutive singular form is based directly on the root. Where a noun has a distinct root for all plural forms, i.e. in weak and strong suppletive nouns, the absolutive singular is formed on the singular root. These facts about Archi are formalised in (1) to (4) below.

In (1) we express the default dependence of a morphological word-form on the lexeme's stem, and in this way formalise number marking through the stem. This is because the path <mor> implies any extension of itself, as does the path <stem>. This expresses the fact that, for example, the path <mor sg> takes as its value the value of the path <stem sg>. The implication is that all singular word forms, regardless of case, will begin with a singular stem. The quotes express that <stem> and its extensions, <stem sg> and <stem pl> and as we shall see later <stem sg

erg>, are retrieved from the lexical entry being queried. Finally, the ‘exception’ is the absolutive singular which is formed directly on a root form: in the case of some items, on the bare root, for others a special form of the root used in the singular only, and for others a special root used for the absolutive singular only. These three possibilities correspond to the sub-path <root>, its extension <root sg> and its further extension <root sg abs>. The quotes indicate that the path is lexically specified.

(1)

NOUN:

```
<mor> == "<stem>"
<mor sg abs> == "<root sg abs>".
```

With respect to the formation of the absolutive and ergative, Archi can be said to have three major nouns classes has three noun classes. These generalise the formation of the stem when it combines with inflections to express the full set of morphosyntactic features. One class is for lexemes with consonant final roots, which we call Class 1. Another is for lexemes with vowel final roots (Class 2). There is also a special class for substantivised adjectives, and these lexemes have frozen gender markers (Class 3). The three stem formation classes are given in (2) and (3) below. In (2a) the first class, Class 1, inherits facts about nouns, including those discussed above, and specifies singular and plural stem building generalisations. The singular stem is based on the lexically specified ergative root, in those instances where this is distinct, and the plural on a plural root, again in instances where there is a distinction in the roots, to which is added the formative *-mul-*.

(2a)

```
CLASS_1:
<> == NOUN
<stem sg> == "<root sg erg>" CLASS_1_SG:<>
<stem pl> == "<root pl>" mul CLASS_1_PL:<>.
```

For singular stem formation, referral is made to a special node (2b) where by default the formative *-li-* is added to the root, and then case endings are picked up at a separate node (discussed below). However, for the absolutive nothing is added to the root, and no reference is made to inflectional material to mark case ending.

(2b)

```
CLASS_1_SG:
<> == li NOUN_FORMS:<>
<abs> == .
```

Plural stem formation is expressed similarly with the important difference that the formative *-čaj-* is added to all stems; the exception is the absolutive marker which is zero (2c).

(2c)

```
CLASS_1_PL:  
  <> == čaj NOUN_FORMS:<>  
  <abs> == .
```

Singular and plural stem formation operates differently in the other two classes, and this is shown in (3). Note that Class 2 operates in the same way as Class 1 for singular stem formation, and this is expressed by using the empty path.

(3)

```
CLASS_2:  
  <> == CLASS_1  
  <stem pl> == "<root pl>" CLASS_2_PL:<>.
```

```
CLASS_2_PL:  
  <> == t:aj NOUN_FORMS:<>  
  <abs> == t:u.
```

```
CLASS_3:  
  <> == NOUN  
  <stem sg> == "<root sg erg>" CLASS_3_SG:<>  
  <stem pl> == "<root pl>" CLASS_3_PL:<>.
```

```
CLASS_3_SG:  
  <> == mu NOUN_FORMS:<>.
```

```
CLASS_3_PL:  
  <> == maj NOUN_FORMS:<>  
  <abs> == .
```

Each stem is ultimately referred to a list of case features and corresponding inflections, shown in (4). As word-forms are based on the ergative word-form, the ergative receives no additional marker.

(4)

```
NOUN_FORMS:  
  <erg> ==  
  <gen> == n  
  <dat> == s  
  <comit> == ʈ:u  
  <comp> == xur  
  <perm> == kʈ'əna  
  <part> == qʈiš  
  <superlat> == t:ik  
  <sublat> == kʈ'ak.
```

### 3 Regular Lexical entries

Having provided a formal account of the noun system, we can now show how lexical entries inherit these facts. Lexical entries are represented as nodes which are labeled to indicate the lexeme being represented. Each lexical entry is furnished with a root. For some, a special version of the root is used for building the oblique singular stems and another for building plural stems. This is true for *Kʰánnu* ‘lover’. Lexical entries inherit from one of the three stem formation classes. This is shown in (5a-c).

(5a)

Árum:

```
<> == CLASS_1
<gloss> == sickle
<root> == áˈrum .
```

(5b)

Á<sup>ˈ</sup>ri:

```
<> == CLASS_2
<gloss> == military division
<root> == áˈri.
```

(5c)

Kʰánnu:

```
<> == CLASS_3
<gloss> == lover
<root> == kʰánnu
<root sg erg> == <root> m
<root pl> == kʰánnib.
```

### 4 Deponent lexical entries

We mentioned four properties associated with Archi deponent nouns, and we now show how each is captured. We start with the third property, namely that the mismatch between morphosyntactic feature formal expression lies at the level of the stem. This means that the mismatch between form and function is essentially a mismatch between stem shape and the number feature normally identified with it.

The word *ha ʕara* ‘river’ is a deponent noun whose singular morphosyntax is realised by what appear to be plural forms. This is because rather than use the singular stem formation rule for its class, it uses the plural stem formation rule. The lexical entry is given in (6a) where the last line shows how this kind of deponency is captured as a referral to the plural formation process in order to build the singular stem.

(6a)

Ha<sup>ʔ</sup>təra:  
<> == CLASS\_1  
<gloss> == river  
<root> == ha<sup>ʔ</sup>tər  
<root sg abs> == ha<sup>ʔ</sup>təra  
<stem sg> == <root sg> CLASS\_1\_PL:<>.

The theorem of the lexical entry Ha<sup>ʔ</sup>təra is given in (6b). It clearly shows the plural formative *-čaj-* (see 1c above). The theorem also shows the fourth property of Archi deponents, the fact that the root is not involved in the mismatch: in (5a) we specifically say that stem formation will be plural for the singular stem, but that it will be built on a singular root. This is important for weak suppletive nouns which have distinct singular and plural roots, as we shall see.

(6b)

Ha<sup>ʔ</sup>təra:<gloss> = river.  
Ha<sup>ʔ</sup>təra:<mor sg abs> = ha<sup>ʔ</sup>təra.  
Ha<sup>ʔ</sup>təra:<mor sg erg> = ha<sup>ʔ</sup>tər čaj.  
Ha<sup>ʔ</sup>təra:<mor sg gen> = ha<sup>ʔ</sup>tər čaj n.  
Ha<sup>ʔ</sup>təra:<mor sg dat> = ha<sup>ʔ</sup>tər čaj s.  
Ha<sup>ʔ</sup>təra:<mor sg comit> = ha<sup>ʔ</sup>tər čaj ʔ:u.  
Ha<sup>ʔ</sup>təra:<mor sg comp> = ha<sup>ʔ</sup>tər čaj xur.  
Ha<sup>ʔ</sup>təra:<mor sg perm> = ha<sup>ʔ</sup>tər čaj kʔəna.  
Ha<sup>ʔ</sup>təra:<mor sg part> = ha<sup>ʔ</sup>tər čaj qʔiš.  
Ha<sup>ʔ</sup>təra:<mor sg superlat> = ha<sup>ʔ</sup>tər čaj t:ik.  
Ha<sup>ʔ</sup>təra:<mor sg sublat> = ha<sup>ʔ</sup>tər čaj kʔak.  
Ha<sup>ʔ</sup>təra:<mor pl abs> = ha<sup>ʔ</sup>tər mul.  
Ha<sup>ʔ</sup>təra:<mor pl erg> = ha<sup>ʔ</sup>tər mul čaj.  
Ha<sup>ʔ</sup>təra:<mor pl gen> = ha<sup>ʔ</sup>tər mul čaj n.  
Ha<sup>ʔ</sup>təra:<mor pl dat> = ha<sup>ʔ</sup>tər mul čaj s.  
Ha<sup>ʔ</sup>təra:<mor pl comit> = ha<sup>ʔ</sup>tər mul čaj ʔ:u.  
Ha<sup>ʔ</sup>təra:<mor pl comp> = ha<sup>ʔ</sup>tər mul čaj xur.  
Ha<sup>ʔ</sup>təra:<mor pl perm> = ha<sup>ʔ</sup>tər mul čaj kʔəna.  
Ha<sup>ʔ</sup>təra:<mor pl part> = ha<sup>ʔ</sup>tər mul čaj qʔiš.  
Ha<sup>ʔ</sup>təra:<mor pl superlat> = ha<sup>ʔ</sup>tər mul čaj t:ik.  
Ha<sup>ʔ</sup>təra:<mor pl sublat> = ha<sup>ʔ</sup>tər mul čaj kʔak.

The theorem in (6b) also shows the first property of Archi deponent nouns mentioned in section 1: unlike Latin there is no defectiveness in the paradigm correlating with the mismatch. So we see clearly that Ha<sup>ʔ</sup>təra's theorem has a full set of singular and plural forms. This is due to the fourth property, that the expression of number on the root is not involved in the mismatch. In our example the plural morphosyntax is realised in the same way as for any other class 1 noun where a formal number distinction can be made because the singular is based directly on the singular root, but the plural is based on a root that has been extended by the *-mul-* formative in accordance with the class.

To sum up, the analysis so far captures three of the properties of Archi deponency. First, that there is no defectiveness in a deponent nouns paradigm (property 1), second the mismatch involves number marking by the stem (property 3). And third, that the root is not involved in the mismatch, only the stem building rules (property 4). This leaves one final property to address, property 2: deponency in Archi involves heteroclis. The example lexical entries in (7) and (8) show how property 2 is captured. As these items also have distinct plural roots, they demonstrate more explicitly the fourth property, that the root is not involved in the mismatch: singular stems are built on singular roots, plural stems are built on plural roots

(7a)

C'aj:  
 <> == NOUN  
 <gloss> == goat  
 <root> == c'aj  
 <root pl> == c'ohor  
 <stem sg> == <root sg erg> CLASS\_2\_PL:<>  
 <stem pl> == <root pl> CLASS\_1\_PL:<>.

(7b)

C'aj:<gloss> = goat.  
 C'aj:<mor sg abs> = c'aj.  
 C'aj:<mor sg erg> = c'aj t:aj.  
 C'aj:<mor sg gen> = c'aj t:aj n.  
 C'aj:<mor sg dat> = c'aj t:aj s.  
 C'aj:<mor sg comit> = c'aj t:aj ʈ:u.  
 C'aj:<mor sg comp> = c'aj t:aj xur.  
 C'aj:<mor sg perm> = c'aj t:aj kʈ'əna.  
 C'aj:<mor sg part> = c'aj t:aj q'iš.  
 C'aj:<mor sg superlat> = c'aj t:aj t:ik.  
 C'aj:<mor sg sublat> = c'aj t:aj kʈ'ak.  
 C'aj:<mor pl abs> = c'ohor.  
 C'aj:<mor pl erg> = c'ohor čaj.  
 C'aj:<mor pl gen> = c'ohor čaj n.  
 C'aj:<mor pl dat> = c'ohor čaj s.  
 C'aj:<mor pl comit> = c'ohor čaj ʈ:u.  
 C'aj:<mor pl comp> = c'ohor čaj xur.  
 C'aj:<mor pl perm> = c'ohor čaj kʈ'əna.  
 C'aj:<mor pl part> = c'ohor čaj q'iš.  
 C'aj:<mor pl superlat> = c'ohor čaj t:ik.  
 C'aj:<mor pl sublat> = c'ohor čaj kʈ'ak.

(7a) is the lexical entry for *c'aj* 'goat' and (7b) shows its theorem. (7a) shows that singular stem formation takes place at the Class 2 plural formation node. However, it is a singular root that is involved, specifically the ergative (or oblique) singular root. This is defined as *c'ej* in contrast to the plural root defined as *c'ohor*. Note that this

item has three root versions when we count the default root *c'aj*. Heterocclisis is seen in the fact that singular stem formation is referred to a Class 2 stem formation node, and plural stem formation to a Class 1 node. The lexical entry for *xali* 'family' is treated in the same way (8a, b). In this case the heterocclisis involves Class 3 and Class 2. Unlike *c'aj* 'goat' there is no distinct plural root, but there is a distinct ergative singular root, a truncated version of the default root: *xal*.

(8a)

Xali:  
 <> == NOUN  
 <gloss> == family  
 <root> == xali  
 <root sg erg> == xal  
 <stem sg> == <root sg erg> CLASS\_3\_PL:<>  
 <stem pl> == <root pl> CLASS\_2\_PL:<>.

(8b)

Xali:<gloss> = family.  
 Xali:<mor sg abs> = xali.  
 Xali:<mor sg erg> = xal maj.  
 Xali:<mor sg gen> = xal maj n.  
 Xali:<mor sg dat> = xal maj s.  
 Xali:<mor sg comit> = xal maj ɬ:u.  
 Xali:<mor sg comp> = xal maj xur.  
 Xali:<mor sg perm> = xal maj kɬ'əna.  
 Xali:<mor sg part> = xal maj q'iš.  
 Xali:<mor sg superlat> = xal maj t:ik.  
 Xali:<mor sg sublat> = xal maj kɬ'ak.  
 Xali:<mor pl abs> = xali t:u.  
 Xali:<mor pl erg> = xali t:aj.  
 Xali:<mor pl gen> = xali t:aj n.  
 Xali:<mor pl dat> = xali t:aj s.  
 Xali:<mor pl comit> = xali t:aj ɬ:u.  
 Xali:<mor pl comp> = xali t:aj xur.  
 Xali:<mor pl perm> = xali t:aj kɬ'əna.  
 Xali:<mor pl part> = xali t:aj q'iš.  
 Xali:<mor pl superlat> = xali t:aj t:ik.  
 Xali:<mor pl sublat> = xali t:aj kɬ'ak.

## 5 Suppletive lexical entries

There is one deponent noun which is also suppletive, the word *x'on* 'cow' which has the root *buc:'i* for the plural. For plural morphosyntax it uses a singular stem formation pattern, that of Class 1. It can be both suppletive and deponent because of the fourth property, the mismatch does not involve roots: for singular morphosyntax

one root is used, and for plural the other. Note that for this item the deponency involves a referral to Class 1 singular stem formation to build the plural stem, but for the singular stem no stem formation class is involved: singular forms are built directly on the ergative singular root, *x<sup>ɸ</sup>ini*. The absolutive singular is built on the (singular) root by default, as shown in (1) above.

(9a)

```
Xɸon:
  <> == NOUN
  <gloss> == cow
  <root sg abs> == xɸon
  <root sg erg> == xɸini
  <root pl> == buc:'i
  <stem sg> == <root sg erg> NOUN_FORMS:<>
  <stem pl> == <root pl> CLASS_1_SG:<>.
```

(9b)

```
Xɸon:<gloss> = cow.
Xɸon:<mor sg abs> = xɸon.
Xɸon:<mor sg erg> = xɸini.
Xɸon:<mor sg gen> = xɸini n.
Xɸon:<mor sg dat> = xɸini s.
Xɸon:<mor sg comit> = xɸini ɬ:u.
Xɸon:<mor sg comp> = xɸini xur.
Xɸon:<mor sg perm> = xɸini kɬ'əna.
Xɸon:<mor sg part> = xɸini q'iš.
Xɸon:<mor sg superlat> = xɸini t:ik.
Xɸon:<mor sg sublat> = xɸini kɬ'ak.
Xɸon:<mor pl abs> = buc:'i.
Xɸon:<mor pl erg> = buc:'i li.
Xɸon:<mor pl gen> = buc:'i li n.
Xɸon:<mor pl dat> = buc:'i li s.
Xɸon:<mor pl comit> = buc:'i li ɬ:u.
Xɸon:<mor pl comp> = buc:'i li xur.
Xɸon:<mor pl perm> = buc:'i li kɬ'əna.
Xɸon:<mor pl part> = buc:'i li q'iš.
Xɸon:<mor pl superlat> = buc:'i li t:ik.
Xɸon:<mor pl sublat> = buc:'i li kɬ'ak.
```

For completeness we include suppletive nouns that are not deponent. The word ‘corner of a sack’ has one root for the singular and another formally distinct root for the plural. The theory accounts for this naturally as it already accounts for a distinction between the singular and plural versions of the root. Note that for plural indirect cases there is an epenthetic /r/, e.g. ergative plural *boždorčaj*. This is not accounted for.



(10a)

Bič'ni:

<> == CLASS\_1  
<gloss> == corner of sack  
<root sg> == bič'ni  
<root pl> == boždo  
<stem pl> == <root pl> CLASS\_1\_PL:<>.

(10b)

Bič'ni:<gloss> = corner of sack.  
Bič'ni:<mor sg abs> = bič'ni.  
Bič'ni:<mor sg erg> = bič'ni li.  
Bič'ni:<mor sg gen> = bič'ni li n.  
Bič'ni:<mor sg dat> = bič'ni li s.  
Bič'ni:<mor sg comit> = bič'ni li ʈ:u.  
Bič'ni:<mor sg comp> = bič'ni li xur.  
Bič'ni:<mor sg perm> = bič'ni li kʈ'əna.  
Bič'ni:<mor sg part> = bič'ni li q'iš.  
Bič'ni:<mor sg superlat> = bič'ni li t:ik.  
Bič'ni:<mor sg sublat> = bič'ni li kʈ'ak.  
Bič'ni:<mor pl abs> = boždo.  
Bič'ni:<mor pl erg> = boždo čaj.  
Bič'ni:<mor pl gen> = boždo čaj n.  
Bič'ni:<mor pl dat> = boždo čaj s.  
Bič'ni:<mor pl comit> = boždo čaj ʈ:u.  
Bič'ni:<mor pl comp> = boždo čaj xur.  
Bič'ni:<mor pl perm> = boždo čaj kʈ'əna.  
Bič'ni:<mor pl part> = boždo čaj q'iš.  
Bič'ni:<mor pl superlat> = boždo čaj t:ik.  
Bič'ni:<mor pl sublat> = boždo čaj kʈ'ak.

In our analysis we also provide for a distinct ergative singular root. This means we can handle the suppletive noun 'father' which is defective for the plural, but has formally distinct stems for the absolutive singular and all other case forms. The lexical entry for this item is shown in (11a).

(11a)

Abt:u:

<> == CLASS\_3  
<gloss> == father  
<root sg abs> == abt:u  
<root sg erg> == um.

The theorem is given in (11b). Note that because the plural stem is nowhere defined in the lexical entry, nor can it be inferred from the root as this is not defined either, the

plural is undefined, i.e. defective. In other words, the specificity of the suppletion involved prevents plural formation.

(11b)

- Abt:u:<gloss> = father.
- Abt:u:<mor sg abs> = abt:u.
- Abt:u:<mor sg erg> = um mu.
- Abt:u:<mor sg gen> = um mu n.
- Abt:u:<mor sg dat> = um mu s.
- Abt:u:<mor sg comit> = um mu ʃ:u.
- Abt:u:<mor sg comp> = um mu xur.
- Abt:u:<mor sg perm> = um mu kʰəna.
- Abt:u:<mor sg part> = um mu qʰiʃ.
- Abt:u:<mor sg superlat> = um mu t:ik.
- Abt:u:<mor sg sublat> = um mu kʰak.

## 6 Conclusions

We can distinguish four properties associated with deponency in Archi. There is no accompanying defectiveness (property 1), but heteroclisis is involved (property 2). A possible explanation for property 1 lies in property 2: the heteroclisis allows a number distinction to be maintained. Two further properties may also explain property 1: number is not marked by the inflection, rather the stem (property 3), and the root version (property 4). Only the stem is involved in the feature ~ form mismatch, so a distinction can be made not only through heteroclisis, but also through the form of the root. We have shown how these properties and the implications they carry for describing the facts of deponency in Archi can be subjected to a formal treatment.

## References

- Baerman, Matthew. 2006. Archi Language Report [available at <http://www.smg.surrey.ac.uk/deponency/Examples/Archi.htm>]
- Corbett, Greville G.; and Fraser, Norman. 1993. Network Morphology: a DATR account of Russian nominal inflection. *Journal of Linguistics* 29 (1). 113-42.
- Evans, Nicholas; Brown, Dunstan; and Corbett, Greville G. 2001. Dalobon pronominal prefixes and the typology of syncretism. *Yearbook of Morphology 2000*. Dordrecht: Kluwer: 187-231.
- Evans, Roger; and Gazdar, Gerald. 1996. DATR: a language for lexical knowledge representation. *Computational Linguistics* 22 (2).167-216.
- Kibrik, Aleksandr E. 1977a. *Opyt strukturnogo opisanija arčinskogo jazyka: II: Taksonomičeskaja grammatika*. (Publikacii otdelenija strukturnoj i prikladnoj lingvistiki 12). Moscow: Izdatel'stvo Moskovskogo universiteta
- Kibrik, Aleksandr Evgenievič. 1977b. *Arčinskij jazyk: teksty i slovar'*. Moscow: Izdatel'stvo Moskovskogo Universiteta.
- Hippisley, Andrew. 2001. Word Formation Rules in a default inheritance framework: a Network Account of Russian personal nouns. In: Jaap van Marle and Geert Booij (eds) *Yearbook of Morphology 1999*. Dordrecht: Kluwer.221-261
- Hippisley, Andrew; Chumakina, Marina; Corbett, Greville and Dunstan Brown 2004. Suppletion: frequency, categories and distribution of stems. *Studies in Language* 28 (2).389-421.